Part of this document is from SAS/IML User's Guide.
· What is IML (Interactive Matrix Language)? A powerful and flexible programming language in a dynamic, interactive environment.

1. SAS/IML software is a programming language. It has many built-in operators and routines, a complete set of control statements, and easy access to SAS datasets
2. SAS/IML software operates on matrices. 
3. SAS/IML software possesses a powerful vocabulary of operators.
4. SAS/IML software uses operators that apply to entire matrices.
5. SAS/IML software is interactive.
6. SAS/IML software is dynamic.
7. SAS/IML software processes data.
8. SAS/IML software produces graphics.
· Outline: we will introduce matrix, matrix operations, control statements, functions or subroutines, and some examples.
options ls=80 nodate;

proc iml;

a = {1 5 8, 5 3 6, 8 6 4};

print a;

b = { 5 2 4, 1 3 2, -5 6 7};

print b;

c = a + b;

quit;  
The quit statement or a new SAS procedure/data step end the iml session. One cannot use the run statement to end the iml session. The run statement has special meanings in an iml session. The iml program executes statement by statement
proc means data = a;

run;  
The matrix “a” created during the above iml session is not recognized outside of the defining iml session
proc iml;/* this starts iml */
/* Create a 3 by 3 matrix called a. Spaces separate entries within a row, a comma separates rows. */
a = {1 5 8, 5 3 6, 8 6 4};

b = { 5 2 4, 1 3 2, -5 6 7};

imat = I(5); /* I(k) create a k by k identity matrix */
/* J(nrow,ncol,value) creates a matrix with nrow rows, ncol columns and value for each entry */
c = J(3,4,2); /* creates a 3 by 4 matrix of 2's */
d = J(2,3); /* no value given. The default is 1 */
e = {1 1 5 6 7 9 11 11}; /* creates a 1 by 8 row vector */
/* row vectors can be created using the index operator :

r = i:j will create a row vector going from i up to j by steps of 1 if j is bigger than i or going from i down to j if i is bigger than j. r = do(l,u,inc) will create a row vector going from l to u by steps of inc */
r1 = 1:5;

r2 = do(10,18,2);

print a b imat c d e r1 r2;

n= nrow(c); /* nrow() finds number of rows. */
p = ncol(c); /* ncol() finds number of columns */
print "number of rows in c is " n "number of columns in c is" p;

print "number of rows in c is " n [format=1.] ",    number of columns in c is" p;

/*Syntax for print:

print <matrix> <(expression)> <"message"> <[options]>;

options: 

format = 

colname=

rowname=

*/
print (a[1,1]);

/* basic matrix operations */
sumab = a + b;

diffab = a - b;

ratioab=a/b; /* elementwise ratio */
prodab = a*b; /* multiplication */
eprodab=a#b; /* elementwise multiplication*/
print prodab eprodab;

acubed = a**3; /* matrix to a power a*a*a* */
a3 = a##3; /* elementwise power */
apb =a ##b; /* elementwise power with two matrices*/
tr1 = t(r1); /* t() finds the transpose. */
q = r2*t(r1); /* same as using r2*tr1 with tr1 defined earlier*/
print sumab diffab ratioab prodab eprodab acubed a3 apb tr1 q;

/*For elementwise operations, a and b must have the same dimension, or one of them is a constant*/
/* SOME MATRIX FUNCTIONS */
sumr1 = sum(r1); /* sum() sums elements of the matrix */
ssqr1= ssq(r1); /* ssq() gets sum of squares of elements */
detb= det(b); /* determinant */
invb=inv(b); /* inverse of nonsingular matrix */
tra=trace(b); /* trace = sum of diagonal elements */
evala = eigval(a); /* eigenvalues = characteristic roots of sym. matrix*/
eigvec= eigvec(a); /* eigenvectors = characteristic vectors of sym. matrix*/
maxa= max(a); /* maximum element */
minr1 = min(r1); /* minimum element */
ranke=rank(e); /* ranks */
ranket=ranktie(e); /* gets ranks, averaging ties */
sqrta=sqrt(a); /* gets square root element by element*/
print sumr1 ssqr1 detb invb tra evala eigvec maxa

minr1 ranke ranket sqrta;

quit;

/*Naming rows and columns, formats, reset*/
options ls=80 nodate;

proc iml; 

a = {1 5 8, 5 3 6, 8 6 4}; 

b = {5 2 4, 1 3 2};

rnames = {mon tue wed};

cnames = {group1 group2 group3};

print a;

print a[rowname=rnames colname=cnames];

mattrib a rowname = (rnames) colname=(cnames);

reset autoname spaces=13; /*spaces = n puts n spaces between columns*/
print a;

print b;

reset noautoname;

print a;

print b;

quit;

/*Referring to elements*/
options ls=80 nodate;

proc iml;

a = {1 5 8, 5 3 6, 8 6 4};

q1 = a[3,1]; /* specifies in row, column form */
q2 = a[5]; /* looks for 5th element of a in row-major form. That is it stretches elements of a out using first row then second row, etc.*/
print q1 q2;

/* selecting row or columns */
r1 = a[1,]; /* r1 is first row of a */
c2 = a[,2]; /* c2 is second column of a*/
x = rank(a[2,]); /* x contains ranks of 2nd row of a */
z = a[1,]*a[,3]; /* mutiplies 1st row of a times 3rd column of a*/
print r1 c2 x z;

/* creating submatrices*/
new = a[{1 3},{2 3}]; /* new consist of rows 1 and 3 and columns 2 and 3 of a*/
new2 = a[1:2,2:3]; /* can use index operator to choose successive rows or columns. new2 has rows 1 and 2, columns 2 and 3 */
print new new2;

quit;

/*Reading from a SAS data set into matrices.*/
options ps=60 ls=80;

data a;

infile 'C:\Documents and Settings\anna\Desktop\MyDesktop\597\speed.dat';

input id sex vo35 vo4 vo45 vo5 vo55 vo6;

proc print;

run;

proc iml;

use a;

read all into bigmat;

read all var {id};

read all var {id sex} into b;

read all var{vo35 vo4 vo45 vo5 vo55 vo6} where(sex=1) into vmat1;

read all var{vo35 vo4 vo45 vo5 vo55 vo6} where(sex=2) into vmat2;

read point 3 var {id} into id1;

read next 5 var {id} into id2;

close a;

print bigmat; print id; print b; print vmat1;

print vmat2; print id1; print id2; 

quit; 

run;

/*Descriptive statistics etc. in a one-sample problem*/
title 'Illustrating descriptive statistics with the LA heart data';

options ls=80 ps=60 nodate;

data values;

infile 'C:\Documents and Settings\anna\Desktop\MyDesktop\597\ladata.dat';

input id age md50 sp50 dp50 ht50 wt50 sc50

soec cs md62 sp62 dp62 sc62 wt62 ihdx yrdth;

run;

proc iml;

use values;

read all var{wt50} into x;

close values;

print x;

n = nrow(x);

/* get mean, variance and standard deviation */
mean = sum(x)/n;

var = (ssq(x) - (n*(mean**2)))/(n-1);

sd=sqrt(var);

reset noname; /* name of matrices will not be printed*/
print "n = " n "mean: " mean "st. deviation: " sd "variance:" var;

/* get 95% confidence interval for the mean and variance */
a=.05;

se = sd/sqrt(n);

tval=tinv(1-a/2,n-1);

upper = mean + tval*se;

lower = mean - tval*se;

print "standard error = " se ;

print "95% CI for mean = " lower " to " upper;

lval = cinv((a/2),n-1);

uval = cinv(1-a/2,n-1);

lowvar = (n-1)*var/uval;

upvar = (n-1)*var/lval;

lowsd=sqrt(lowvar);

upsd=sqrt(upvar);

print "95% CI for variance = " lowvar " to " upvar;

print "95% CI for standard dev. = " lowsd " to " upsd;

/* Organizing results in a matrix for better looking output*/
print "Estimates and 95% confidence intervals";

cimat=J(3,3);

cimat[1,1]=mean;

cimat[1,2]=lower;

cimat[1,3]=upper;

cimat[2,1]=var;

cimat[2,2]=lowvar;

cimat[2,3]=upper;

cimat[3,1]=sd;

cimat[3,2]=lowsd;

cimat[3,3]=upsd;

rname={Mean Variance StDeviation};

cname={Estimate LowerCI UpperCI};

reset noname;

print cimat [rowname=rname colname=cname format=6.2];

/* First create a ranked version of x */
rx=J(n,1); /* initializes rx */
rx[rank(x)]=x; /* rx has the ranked values */
reset name;

print rx x;

/* get quantiles using definition 3 in SAS proc univariate*/
per={1,5,25,50,75,95,99}; /* specify percentiles of interest */
q=per;

k=nrow(per);

do i = 1 to k;

  p=per[i]/100;

  j = int(n*p);

  g = n*p - j;

  if g=0 then q[i]=rx[j];

  else q[i] = rx[j+1];

end;

print per q;

quit;

proc univariate data=values PCTLDEF=3;

var wt50;

run;

/*Base SAS Functions Accessible from SAS/IML

http://support.sas.com/documentation/cdl/en/imlug/59656/HTML/default/langref_sect310.htm*/
/* Creating a chi-square goodness of fit test of the normal

using 10 groups based on the 10th 20th,..., 90th percentile */
data values;

infile 'C:\Documents and Settings\anna\Desktop\MyDesktop\597\ladata.dat';

input id age md50 sp50 dp50 ht50 wt50 sc50

soec cs md62 sp62 dp62 sc62 wt62 ihdx yrdth;

run;

proc iml;

use values;

read all var{wt50} into x;

close values;

n = nrow(x);

rx = J(n, 1);

rx[rank(x)]=x; /* rx has the sorted values */
d=J(9,1); /* initialize a 9 by 1 vector */
do i = 1 to 9;

p=i/10;

j = int(n*p);

g = n*p - j;

if g=0 then d[i]=rx[j];

else d[i] = rx[j+1];

end;

/*The above do-loop can be replaced by

p = (1:9)/100;

g = (n*p - int(n*p)) > 0 ;

d = rx[int(n*p)+ g];

print p g d;

*/
/* get mean, variance and standard deviation */
mean = sum(x)/n;

var = (ssq(x) - (n*(mean**2)))/(n-1);

sd=sqrt(var);

print mean sd;

/* get expected number in each group*/
exp=J(10,1);

exp[1] = cdf('normal',d[1],mean,sd)*n;

do i = 2 to 9;

exp[i] = (cdf('normal',d[i],mean,sd) - cdf('normal',d[i-1],mean,sd))*n;

end;

exp[10] = (1- cdf('normal',d[9],mean,sd))*n;

chisq=0;

obs=exp;

do i = 1 to 10;

obs[i]=.1*n;

chisq=chisq + ((obs[i] - exp[i])**2)/exp[i];

end;

/*The previous do-loop can be replaced by  

chisq = sum((obs-exp)##2/exp); */
pvalue = 1 - probchi(chisq,7);

reset spaces=4;

print "chi-square goodness of fit statistic" chisq "pvalue" pvalue;

print d obs exp [format=5.2]; 

/*Creating a SAS data set from a list of variables.*/
create mat var{d obs exp};

append var{d obs exp};

/*To create a SAS data set from a matrix use create sasdataset from matrixname; append from matrixname;*/
/*Writing to external files in IML.*/
do i = 1 to 9;

file 'C:\Documents and Settings\anna\Desktop\MyDesktop\597\goodfit.dat';

put (d[i]) +1 (obs[i]) +1 (exp[i]) 5.2; /* need to put elements in parentheses*/
end;

put @6 (obs[10]) +1 (exp[10]) 5.2;

      quit;
/* Least square regression via IML*/
/*Example.

Story Names: Smoking and Cancer

Reference: J.F. Fraumeni, "Cigarette Smoking and Cancers of the Urinary Tract: Geographic Variations Authorization: free use

The data are per capita numbers of cigarettes smoked (sold) by 43 states and the District of Columbia in 1960 together with death rates per thouusand population from various forms of cancer.

Variable Names:

1. CIG = Number of cigarettes smoked (hds per capita)

2. BLAD = Deaths per 100K population from bladder cancer

3. LUNG = Deathes per 100K population from lung cancer

4. KID = Deaths per 100K population from kidney cancer

5. LEUK = Deaths per 100 K population from leukemia

*/
option ls=80 nodate;

data a;

infile 'C:\Documents and Settings\anna\Desktop\MyDesktop\597\cig.dat';

input state $ cig blad lung kid leuk;

proc print;

run;

proc gplot data=a;

plot (blad lung kid leuk)*cig;

run;

/* least squares linear fit via proc reg */
proc reg;

model lung=cig;

plot lung*cig p.*cig/overlay;

plot rstudent.*cig;

run;

/* doing linear least squares in IML */
proc iml;

use a;

read all var{lung} into y;

read all var {cig} into x;

close a;

n=nrow(x);

Xmat = J(n, 2);

Xmat[,2] = x;

betahat = inv(t(Xmat)*Xmat)*t(Xmat)*y;

cig=x;

yhat = Xmat*betahat;

create fit var {cig yhat y};

append;

close fit;

/*Create a least square regression module*/
start reg(x1, y) global(betahat1);

n=nrow(x1);

Xmat = J(n,2);

Xmat[,2]=x1;

betahat1 = inv(t(Xmat)*Xmat)*t(Xmat)*y;

yhat1 = Xmat*betahat1;

return(yhat1);

finish reg;

/*Run the regression module for log(y) versus log(cig)*/
fit=reg(log(x), log(y));

yhat1=exp(fit);

create fit var {cig yhat y yhat1};

append;

close fit;

/*Use a built-in nonlinear regression module NLPLM (NonLinear Problem with Levenberg-Marquardt method)

to fit model y=beta1*(cig-beta3)^beta2+epsilon*/
start FUN(b) global(n, x, y);

f=y-b[1]*(x-b[3])##b[2];

return(f);

finish FUN;

/* for nlplm the first component of opt must be the number

of observations. The second component controls the printing.

Setting it equal to 0 as here, leads to no automatic output */
opt = J(2, 1, 0);

opt[1] = n;

bini = J(3,1, 0);

bini[1]=0;

bini[2]=0;

call nlplm(rc,bres,"FUN",bini,opt);

print rc bres bini;

yhat2=bres[1]*(x-bres[3])##bres[2];

create fit var {cig yhat y yhat1 yhat2};

append;

close fit;

quit;

run;

proc sort data=fit;

by cig;

run;

symbol1 color=blue;

symbol2 color=blue i=j; *interplot=join;
symbol3 color=red i=j;

symbol4 color=green i=j;

proc gplot data=fit;

plot y*cig=1 yhat*cig=2 yhat1*cig=3 yhat2*cig=4/overlay;

run;

/*more on User defined modules/subroutines/functions*/
option linesize=80 pagesize=60 nodate;

data a;

infile 'C:\Documents and Settings\anna\Desktop\MyDesktop\597\nut2.dat';

input id method lnrsty k1-k3 a1-a3 b1-b2;

changeb=b2-b1;

if method=3 or method=2;

run;

proc iml;

/*This is an example of a subroutine: no return values in the module*/
start meansd(xvec) global(n,mean,var,sd);

n=nrow(xvec);

mean=sum(xvec)/n;

var = (ssq(xvec) - (n*(mean**2)))/(n-1);

sd=sqrt(var);

finish meansd;

use a;

read all var{changeb} where(method=2 & changeb ^=.) into met2;

read all var{changeb} where(method=3 & changeb ^=.) into met3;

close a;

print met2 met3;

/* getting statistics for each group using separate vectors*/
print "group1 = method 2, group2 = method 3 ";

call meansd(met2);

print "group1 = method2", n mean var sd;

call meansd(met3);

print "group2 = method3", n mean var sd;

quit;

